

GB004: 2次元熱伝導

本 whitepaper は Gunnar Backstrom 氏の承諾のもと、書籍“*Simple Fields of Physics by Finite Element Analysis*” に記されている多数の FlexPDE 適用事例の中からその一部を紹介するものです。

オブジェクトが熱的に定常的な状態にあるということは、任意の体積要素から出て行く熱量とその中で生成される熱量とが等しいことを意味します。これを数学的に表現すると次のようになります。

$$\nabla \cdot \mathbf{f} = h(x, y) \quad (1)$$

ここに f は

$$\mathbf{f} = -\lambda \nabla T \quad (2)$$

によって定義され、熱流束密度 (heat flux density、単位面積当りに伝達される熱量) を表します。ただし熱伝導率 (thermal conductivity) λ 、 T は絶対温度、 h は単位体積当りに発生する熱量を意味します。この場合の熱源としては材質中を流れる電流、電磁波、化学反応等が考えられます。(1), (2) より支配方程式は次のような 2 階の偏微分方程式となります。

$$\nabla \cdot (-\lambda(x, y) \nabla T) = h(x, y) \quad (3)$$

GB003 で扱った電導の問題とは次のような対応関係があります。


電導	熱伝導
電位 U	温度 T
電場 $\mathbf{E} = -\nabla U$	温度勾配 ∇T
電導率 σ	熱伝導率 λ
電流密度 $\mathbf{J} = \sigma \mathbf{E}$	熱流束密度 $\mathbf{f} = -\lambda \nabla T$

なお、以下に示す用例においては SI 単位系 (m, K, W) が前提となっている点にご注意ください。

1 Case1 - 2本の温水チューブ

右の図は中に2本の管を含む直径 $1m$ の円筒状のパイプの断面図です。左右の管の中にはそれぞれ $50^{\circ}C$ 、 $80^{\circ}C$ の温水が流れているとしたとき、断熱性素材からなる灰色を施した部分における熱伝導の様子を FlexPDE を使って調べてみることにします。ただしパイプは $0^{\circ}C$ の土壌中に埋設されているものとしてします。

以下では温度分布や熱流束密度ベクトル場のプロットと共に、熱流の出入りの帳尻が合っているかどうかを検証してみます。なお、(3) では温度 T に関する2階の方程式が規定されているわけですが、スクリプト作成時には T や t という変数名を使わないようにしてください。 t は時間変数を表す予約語です。



FlexPDE では大文字、小文字は区別されません。

1.1 Problem descriptor [econduction01a.pde]

まずタイトルを設定します。

```
TITLE
  'Two Insulated Tubes'    { hconduction01a.pde }
```

次に演算精度に関するセレクトをセットします。デフォルトは 0.002 なのですが、ここでは精度を多少高めに設定します。

```
SELECT
  Errlim = 3e-4
```

従属変数を定義します。

```
VARIABLES
  temp          { Temperature }
```

偏微分方程式の定義に先立ち、パラメータ類を定義します。これらは境界の定義、及び境界条件の設定に際して使用されます。解析対象のドメイン上に熱源がないため上記数式中における h は 0、また素材の熱伝導率は 0.03 と設定している点に注意してください。

DEFINITIONS

```
r0 = 0.1  d = 0.15  r1 = 0.5
k = 0.03          { Thermal conductivity }
heat = 0          { Power density }
fluxd_x = -k*dx(temp)  fluxd_y = -k*dy(temp)  { Heat flux density }
fluxd = vector(fluxd_x, fluxd_y)  fluxd_m = magnitude(fluxd)
f_angle = sign(fluxd_y)*arccos(fluxd_x/fluxd_m)/PI*180
```

方程式を定義します。従属変数 `temp` の数式表現になっていませんが、DEFINITIONS セクションの情報に基づき、`temp` に関する 2 階の PDE が自動的に導出されます。

EQUATIONS

```
div(fluxd) = heat      { 2nd order PDE in temp }
```

次に境界の形状と境界条件を設定します。外界との境界条件、及び 2 本の管との境界条件、共に Dirichlet 型であるため Value 文を使用します。

BOUNDARIES

```
Region 1
  Start 'outer' (0, -r1) Value(temp) = 273      { Frozen soil }
  Arc(Center = 0,0) Angle = 360 to Close
  Start 'left' (-d - r0, 0) Value(temp) = 323  { Exclude left hot water tube }
  Arc(Center = -d,0) Angle = 360 to Close
  Start 'right' (d - r0, 0) Value(temp) = 353  { Exclude right hot water tube }
  Arc(Center = d,0) Angle = 360 to Close
```

最後に出力すべき情報を指定します。

PLOTS


```
Grid(x, y)
Contour(temp)  Surface(temp)
Vector(fluxd) norm
Contour(f_angle)
Contour(fluxd_x)  Contour(fluxd_y)  Contour(fluxd_m)
Elevation(normal(fluxd)) on 'outer'
Elevation(-normal(fluxd)) on 'left'  { Outwards from tube }
Elevation(-normal(fluxd)) on 'right'
```

END

1.2 実行結果


(1) Grid(x, y)

FlexPDE によって設定されたグリッドの形状を示しています。




(2) Contour(temp)

解析対象領域（ドメイン）上での等温線図は次のようになります。




(3) Surface(temp)

温度 $T(x, y)$ の曲面の形状をプロットしたものです。




(4) Vector(fluxd) norm

熱流束密度 f のベクトルプロットを示したものです。等温線図とは直交したものとなっています。左の管は熱を発散すると同時に吸収もしている点にご注意ください。




(5) Contour(f_angle)

熱流束密度ベクトル f の偏角に関する等高線図をプロットしたものです。




(6) Contour(fluxd_x)

熱流束密度ベクトル f の x 成分の値に関する等高線図です。




(7) Contour(fluxd_y)

熱流束密度ベクトル f の y 成分の値に関する等高線図です。



(8) Contour(fluxd_m)


熱流束密度ベクトル f の絶対値に関する等高線図です。



次に境界上における熱流束密度ベクトル f の法線成分 f_n の値を Elevation コマンドによりプロットしてみます。グラフの形状もさることながら、プロット下部に出力されるその積分値に意味があります。外界に流出して行く熱量と温水チューブから流入して来る熱量とがバランスしていなくてはならないからです。


(9) Elevation(normal(fluxd)) on 'outer'

外部境界上での熱流束密度ベクトル f の外向き法線成分の値をプロットしたのが次の図です。境界上での積分値は 12.225 という値になっています。




(10) Elevation(-normal(fluxd)) on 'left'

次に左側のチューブ上で熱流束密度ベクトル f の外向き法線成分の値をプロットしてみます。ここで言っている“外向き”というのは解析対象のドメインから見て外向きということなので、normal という演算子に対し負号が付いている点に注意してください。境界上での積分値は -2.806 という値になっています。



(11) Elevation(-normal(fluxd)) on 'right'


同様にして右側のチューブ上で熱流束密度ベクトル f の外向き法線成分の値をプロットしてみます。こちらの場合はすべて流入となるため、関数値はすべてマイナスとなっています。境界上での積分値は -9.415 であり、(10) の -2.806 と合わせると流入分の合計値は 12.221 となります。(9) で求めた外界への流出分 12.225 と比べたときの誤差は非常に小さい (0.04%) と言えます。



2 Case2 - 半円形鋼材からの熱伝導

今度は断面が半円形のステンレス製鋼材（長さは十分に長いとします）に電流を流し様に加熱します。周囲は右図のように一辺が $0.2m$ の花崗岩で囲まれているとし、その外側は氷と水によって $0^\circ C$ に保たれている状態を考えます。このモデルの場合には偏微分方程式中に熱源の項 $h(x, y)$ を含めて考える必要があります。単位体積当り花崗岩中では $h = 0$ 、鋼材中では $h = 1e6$ であるとし、また熱伝導率は花崗岩が $k = 3.5$ 、ステンレス鋼が $k = 45$ という値を仮定します。

スクリプトのベースとしては `hconduction01a.pde` を用いますが、鋼材部分を `exclude` するのではなく別個のリージョンとして扱う点が異なります。



2.1 Problem descriptor [`econduction01b.pde`]

基本形は `hconduction01a.pde` と大きく変わりませんが、ドメイン全体を規定するリージョン 1 の他に鋼材部分を規定するリージョン 'steel' が定義されている点に注意してください。リージョン 1 中で設定したパラメータ値はドメイン全体に適用されますが、リージョン 'steel' 中でもパラメータ値の指定を行っているため、該当リージョン中では後者の値が使用されることとなります。

```
TITLE
  'Heated Semi-Circular Rod'    { hconduction01b.pde }

SELECT
  Errlim = 3e-4

VARIABLES
  temp          { Temperature }
```

DEFINITIONS

```

Lx = 0.1  Ly = 0.1  r0 = 0.05
k          { Thermal conductivity }
heat      { Power per unit volume }
fluxd_x = -k*dx(temp)  fluxd_y = -k*dy(temp)  { Heat flux density }
fluxd = vector(fluxd_x, fluxd_y)  fluxd_m = magnitude(fluxd)

```

EQUATIONS

```

div(fluxd) = heat      { 2nd order PDE in temp }

```

BOUNDARIES

```

Region 1  k = 3.5  heat = 0          { Granite, default }
  Start 'outer' (-Lx, -Ly)  Value(temp) = 273
    Line to (Lx, -Ly) to (Lx, Ly) to (-Lx, Ly) to Close
Region 'steel'  k = 45  heat = 1.0e6  { Steel rod }
  Start 'rod' (-r0, 0)
    Line to (r0, 0)
    Arc to (0, r0) to Close

```

PLOTS

```

Grid(x, y)
Contour(temp)  Surface(temp)
Contour(temp) painted on 'steel'
Vector(fluxd) norm
Contour(fluxd_m) painted
Elevation(normal(fluxd)) on 'rod'
Elevation(normal(fluxd)) on 'outer'
  Report(PI*r0^2/2*1e6) as 'Heating power'

```

```


END

```

2.2 実行結果


(1) Grid(x, y)

図面下部に“Grid#3”と表示されていることからわかるように、所定の精度を得るために2回のグリッド再構成が行われています。



(2) Contour(temp)


ドメイン全体としての等温線図は次のようになります。鋼材部分のみのより詳しい等温線図については(4)を参照ください。



(3) Surface(temp)

温度 $T(x, y)$ の曲面の形状をプロットしたものです。

Heated Semi-Circular Rod


14:01:37 6/27/08
FlexPDE 5.0.22

export01b: Grid#3 p2 Nodes=2214 Cells=1071 RMS Err= 1.4e-4
Integral= 13.66016

(4) Contour(temp) painted on 'steel'

鋼材部分のみのより詳しい等温線図です。


Heated Semi-Circular Rod

14:01:37 6/27/08
FlexPDE 5.0.22

export01b: Grid#3 p2 Nodes=2214 Cells=1071 RMS Err= 1.4e-4
Integral= 1.787485


(5) Vector(fluxd) norm

熱流束密度 f のベクトルプロットを示したものです。温度の最高点から周囲に流れが広がっていく様子が示されています。




(6) Contour(fluxd_m) painted

熱流束密度ベクトル f の絶対値に関する等高線図です。鋼材の角の部分にピークがありそうなことはこの図からも見て取れますが、次の elevation プロットの方がわかりやすいでしょう。




(7) Elevation(normal(fluxd)) on 'rod'

鋼材外周上での熱流束密度ベクトル f の外向き法線成分の値をプロットしたのが次の図です。角の部分にピークが現れています。また境界上での積分値は 3934 という値になっています。



(8) Elevation(normal(fluxd)) on 'outer'

外部境界上での熱流束密度ベクトル f の外向き法線成分の値をプロットしたのが次の図です。境界上での積分値は 3920 という値になっています。



(7) で求めた値とは多少乖離が見られます。これは鋼材の角の部分で関数値がピークとなっているため、積分計算の精度が落ちているためです。ちなみにこの elevation プロットでは

```
Report(PI*r0^2/2*1e6) as 'Heating power'
```

という Report 文を設定しているためプロット下部に

$$\frac{\pi \cdot r_0^2}{2} \cdot 10^6$$

の計算値が“Heating power”という名前で出力されます。これは半円全体で生み出される単位時間当りの熱量の合計値であり、その値 3927 が上記積分の理論値となる点にご注意ください。

3 Case3 - 強制対流による冷却


前の例では外界との境界条件として Dirichlet 型を想定して計算を行いました。今度は Neumann 型の境界条件を設定してみます。すなわち 0°C の水を強制対流 (forced convection) させることで冷却を行うものとし。その場合、境界上では

$$\frac{\partial T}{\partial n} \propto (T - 273)$$

という条件を適用することになります (ニュートンの冷却の法則)。今、比例定数を 50 とすると FlexPDE に対しては Value 文の代わりに

```
Natural(temp) = 50*(temp - 273)
```

という文を指定すれば良いことになります。



3.1 Problem descriptor [econduction01c.pde]

ここでは hconduction01b.pde に対する変分のみを記すことにします。

```
TITLE
  'Heated Semi-Circular Rod, Convection'    { hconduction01c.pde }
...
```


境界条件を Value 文から Natural 文に変更します。

```
BOUNDARIES
Region 1 k = 3.5 heat = 0          { Granite, default }
Start 'outer' (-Lx, -Ly) Natural(temp) = 50*(temp - 273)
Line to (Lx, -Ly) to (Lx, Ly) to (-Lx, Ly) to Close
(Region 'steel' の定義はそのまま残す)
```

PLOTS セクションにおいては外周上における温度の値をプロットするための Elevation 文を追加します。


```
PLOTS
...
Elevation(temp) on 'outer'
...

END
```

3.2 実行結果


(1) Grid(x, y)

ほとんど Case2 と変わりませんが、セルの数は若干少なくて済んでいます。




(2) Contour(temp)

ドメイン全体としての等温線図は次のようになります。Case2 と比べると全体的に温度が高くなっている点に注意してください。特に最高温度は 100° 以上高くなっています。




(3) Surface(temp)

温度 $T(x, y)$ の曲面の形状をプロットしたものです。外周上の温度は 273° よりかなり高くなっています。




(4) Contour(temp) painted on 'steel'

鋼材部分のみのより詳しい等温線図です。




(5) Vector(fluxd) norm

熱流束密度 f のベクトルプロットを示したものです。




(6) Contour(fluxd_m) painted

熱流束密度ベクトル f の絶対値に関する等高線図です。




(7) Elevation(temp) on 'outer'

外周上における温度の値をプロットしたものです。前の例では 273° 一定に保たれていたわけですが、今回の場合には周囲の冷却水の温度より 150° 近くも高くなっている部分があることがわかります。




(8) Elevation(normal(fluxd)) on 'rod'

鋼材外周上での熱流束密度ベクトル f の外向き法線成分の値をプロットしたものです。境界上での積分値は 3937 という値になっています。



(9) Elevation(normal(fluxd)) on 'outer'

外部境界上での熱流束密度ベクトル f の外向き法線成分の値をプロットしたものです。境界上での積分値は 3922 という値になっています。理論値 3927 は Case2 の場合と同一です。



4 Case4 - 非等方的材質中での熱伝導

今度は方程式


$$\nabla \cdot (-\lambda(x, y) \nabla T) = h(x, y)$$

における熱伝導率 λ が x 方向と y 方向で異なる値を持つようなケース、すなわち非等方的 (anisotropic) 材質の場合について考えることにします。このような場合には熱伝導方程式

$$\frac{\partial}{\partial x} \left(-\lambda_x(x, y) \frac{\partial T}{\partial x} \right) + \frac{\partial}{\partial y} \left(-\lambda_y(x, y) \frac{\partial T}{\partial y} \right) = h(x, y) \quad (4)$$

を明示的に FlexPDE に入力する必要があります。

右の図は松の柱の断面図を表しています。繊維の向きが y 軸方向にあるとすると、この材質は y 軸方向には熱が伝わりやすく、 x 軸方向には伝わりにくいという性質を持つことになります。今、図の P の位置に z 軸に沿って (紙面に垂直な方向に) 細い電熱線を張り、そこから流束密度が 1000 の熱が伝わってくるケースを想定します。柱の下部は一定温度 0°C に保たれているものとし、他の境界は熱的に遮断されているものとします。 x 軸方向の電導率を $k_x = 0.14$ 、 y 軸方向の電導率を $k_y = 0.35$ として熱伝導の様子を解析してみましょう。



4.1 Problem descriptor [hconduction01d.pde]

基本形は hconduction01a.pde と大きく変わりません。

```
TITLE
  'Anisotropic Conduction in Wood'    { hconduction01d.pde }

SELECT
  Errlim = 1e-4

VARIABLES
  temp                                { Temperature }
```

DEFINITIONS セクションでは熱伝導率 k_x, k_y を定義します。また電熱線の幅を $2w$ 、 $w = 0.001$ と設定します。

DEFINITIONS

```
L = 0.1 w = 0.001
kx = 0.14 ky = 0.35 { Thermal conductivity of pine wood }
heat = 0 { Power per unit volume }
fluxd_x = -kx*dx(temp) fluxd_y = -ky*dy(temp) { Heat flux density }
fluxd = vector(fluxd_x, fluxd_y) fluxd_m = magnitude(fluxd)
```

偏微分方程式は数式 (4) の形で入力します。

EQUATIONS

```
dx(-kx*dx(temp)) + dy(-ky*dy(temp)) = heat
```

境界条件のうち電熱線の部分については $\text{Natural}(temp) = -1000$ という指定になる点に注意してください。

BOUNDARIES

```
Region 1
Start 'outer' (-L, -L)
Value(temp) = 273 Line to (L, -L)
Natural(temp) = 0 Line to (L, L) to (w, L)
Natural(temp) = -1000 Line to (-w, L) { Heater flux }
Natural(temp) = 0 Line to (-L, L) to Close
```

PLOTS セクションにおいては電熱線の周囲の等温線図を拡大表示するために `zoom` 修飾子を使用しています。

PLOTS


```
Grid(x, y)
Contour(temp) Surface(temp)
Contour(temp) painted zoom(-0.1*L, 0.8*L, 0.2*L, 0.2*L)
Vector(fluxd) norm
```

END

4.2 実行結果


(1) Grid(x, y)

メッシュの再構成が2回行われ (Grid#3) 電熱線周囲のメッシュ密度が高いものとなっています。




(2) Contour(temp)

ドメイン全体としての等温線図は次のようになります。電熱線の近傍では縦方向に長い曲線となっていますが、下辺に近づくにつれほぼ水平なものとなって行きます。




(3) Surface(temp)

温度 $T(x, y)$ の曲面の形状をプロットしたものです。




(4) Contour(temp) painted zoom(-0.1*L, 0.8*L, 0.2*L, 0.2*L)

電熱線の近傍の等温線図を塗りつぶしを施し拡大表示したものです。表示されているのは座標点 $(-0.1L, 0.8L)$ より右に $0.2L$ 、上に $0.2L$ の範囲です。



(5) Vector(fluxd) norm

熱流束密度 f のベクトルプロットを示したものです。絶縁壁に沿っては平行な、一方下辺に対しては垂直な流れが表現されています。



export01d: Grid#3 p2 Nodes=2016 Cells=971 RMS Err= 4.1e-5

